
Fourier Series Illustrations

William O. Bray

Our purpose here is to examine computation of Fourier sine and cosine coefficients and illustrate the pointwise convergence theorem.

All output has been deleted; so enter each Input Cell (white background) and hit Shift+Enter.

Computing Fourier Coefficients

Here are a couple of simple functions defined on the interval $[0, \pi]$.

```
In[1]:= g[x_] := Piecewise[{{1, 0 < x < Pi / 2}, {0, Pi / 2 <= x < Pi}}]
gg[x_] := x
```

Computing the Fourier sine coefficients is easy. For example

```
In[3]:= (2 / Pi) Integrate[g[x] Sin[n x], {x, 0, Pi}]
```

```
Out[3]= 
$$\frac{4 \operatorname{Sin}\left[\frac{n \pi}{4}\right]^2}{n \pi}$$

```

gives a formula for the Fourier sine coefficients of g . For plotting purposes, I find it useful to form tables of such coefficients for a range of the index n . I have also included the code to form the partial sums of the series and for the even and odd extensions of a function.

```
In[4]:= fsincoef[f_, N_] :=
  Table[
    (2 / Pi) Evaluate[Integrate[f[t] Sin[n t], {t, 0, Pi}]], {n, 1, N}]
fcoscoef[f_, N_] :=
  Table[(2 / Pi) Evaluate[Integrate[f[t] Cos[n t], {t, 0, Pi}]],
    {n, 0, N + 1}]
pssin[x_, coef_, N_] := Sum[coef[[n]] Sin[n x], {n, 1, N}]
pscoss[x_, coef_, N_] := coef[[1]] / 2 +
  Sum[coef[[n]] Cos[(n - 1) x], {n, 2, N + 1}]
funodd[fun_, x_] := Piecewise[{{-fun[-x], -Pi < x < 0},
  {fun[x], 0 < x < Pi}}]
funeven[fun_, x_] := Piecewise[{{fun[-x], -Pi < x < 0},
  {fun[x], 0 < x < Pi}}]
```

Sine and Cosine Series for g

For example the following generates the first 20 Fourier cosine coefficients of g .

In[10]:=

```
a = fcoscoef[g, 20]
```

Out[10]=

$$\left\{ 1, \frac{2}{\pi}, 0, -\frac{2}{3\pi}, 0, \frac{2}{5\pi}, 0, -\frac{2}{7\pi}, 0, \frac{2}{9\pi}, 0, \right. \\ \left. -\frac{2}{11\pi}, 0, \frac{2}{13\pi}, 0, -\frac{2}{15\pi}, 0, \frac{2}{17\pi}, 0, -\frac{2}{19\pi}, 0, \frac{2}{21\pi} \right\}$$

We can look at the numeric values with the command:

In[11]:=

```
N[%]
```

Out[11]=


```
{1., 0.63662, 0., -0.212207, 0., 0.127324, 0., -0.0909457,
 0., 0.0707355, 0., -0.0578745, 0., 0.0489708, 0.,
 -0.0424413, 0., 0.0374482, 0., -0.0335063, 0., 0.0303152}
```

The plot of the function along with some partial sums is given as follows.

In[12]:=

```
Plot[{funeven[g, x], pscos[x, a, 10], pscos[x, a, 15]},
  {x, -Pi, Pi}, PlotStyle -> {{Thick, Black}, Blue, {Red, Thick}},
  ImageSize -> {500, 350}]
```

Out[12]=



For the sine series, the commands are similar as follows. In both instances note the behaviour of the partial sums near the jump discontinuity; the Gibbs effect is apparent.

In[13]:=

```
b = fsincoef[g, 20]
```

Out[13]=


$$\left\{ \frac{2}{\pi}, \frac{2}{\pi}, \frac{2}{3\pi}, 0, \frac{2}{5\pi}, \frac{2}{3\pi}, \frac{2}{7\pi}, 0, \frac{2}{9\pi}, \frac{2}{5\pi}, \right.$$

$$\left. \frac{2}{11\pi}, 0, \frac{2}{13\pi}, \frac{2}{7\pi}, \frac{2}{15\pi}, 0, \frac{2}{17\pi}, \frac{2}{9\pi}, \frac{2}{19\pi}, 0 \right\}$$

In[14]:=

```
Plot[{funodd[g, x], pssin[x, b, 10],
      pssin[x, b, 15], pssin[x, b, 20]}, {x, -Pi, Pi},
      PlotStyle -> {{Thick, Black}, Blue, Red, {Orange, Thick}},
      ImageSize -> {500, 350}]
```

Out[14]=



To illustrate convergence of the partial sums more dynamically, the following code does the trick

Fourier Sine and Cosine Series of gg

The following code presents similar computations and graphical illustration for the function gg.

In[17]:=

```
d1 = fcoscoef[gg, 20]
```

Out[17]=


$$\left\{ \pi, -\frac{4}{\pi}, 0, -\frac{4}{9\pi}, 0, -\frac{4}{25\pi}, 0, -\frac{4}{49\pi}, 0, -\frac{4}{81\pi}, 0, -\frac{4}{121\pi}, \right.$$

$$\left. 0, -\frac{4}{169\pi}, 0, -\frac{4}{225\pi}, 0, -\frac{4}{289\pi}, 0, -\frac{4}{361\pi}, 0, -\frac{4}{441\pi} \right\}$$

In[18]:=

```
Plot[{funeven[gg, x], pscos[x, d1, 2],
      pscos[x, d1, 4], pscos[x, d1, 15]},
     {x, -Pi, Pi}, PlotStyle ->
     {{Thick, Black}, Orange, Blue, {Thick, Red}},
     ImageSize -> {500, 350}]
```

Out[18]=



The above plot illustrates uniform convergence of the partial sums to the function.

In[19]:=

```
d2 = fsincoef[gg, 20]
```


Out[19]=

$$\left\{ 2, -1, \frac{2}{3}, -\frac{1}{2}, \frac{2}{5}, -\frac{1}{3}, \frac{2}{7}, -\frac{1}{4}, \frac{2}{9}, -\frac{1}{5}, \right. \\ \left. \frac{2}{11}, -\frac{1}{6}, \frac{2}{13}, -\frac{1}{7}, \frac{2}{15}, -\frac{1}{8}, \frac{2}{17}, -\frac{1}{9}, \frac{2}{19}, -\frac{1}{10} \right\}$$

In[20]:=

```
Plot[{funodd[gg, x], pssin[x, d2, 3],
      pssin[x, d2, 5], pssin[x, d2, 9]},
      {x, -Pi, Pi}, PlotStyle ->
      {{Thick, Black}, Blue, Red, {Thick, Orange}},
      ImageSize -> {500, 350}]
```

Out[20]=



A non-piecewise continuous function

In[21]:=

```
h[x_] := x^(-1/2)
```

This function is continuous everywhere on the interval $[0, \pi]$ except at the point $x = 0$. The Fourier cosine coefficients are given by (here the built in command for the coefficients is used; see the documentation):

In[22]:=

```
FourierCosCoefficient[h[x], x, n]
```

Out[22]=

$$\frac{2 \sqrt{\frac{2}{\pi}} \text{FresnelC}\left[\sqrt{2} \sqrt{\text{Abs}[n]}\right]}{\sqrt{\text{Abs}[n]}}$$

This may cause a "What???" in your mind. The expression on the right is one of the many special functions out there and is known as the Fresnel Cosine integral (see the Documentation Center for a definition). It

originally occurred in various optics problems.

None the less, we can capture the numerical values and plot the results as follows. This is followed by the same computations for the sine series.

In[23]:=

```
c = N[fcoscoef[h, 20]]
```


Out[23]=

```
{2.25676, 0.843989, 0.550935, 0.466778, 0.395447,
0.359079, 0.324163, 0.30273, 0.281207, 0.266664, 0.251744,
0.241042, 0.229933, 0.221631, 0.212953, 0.206267,
0.199248, 0.193713, 0.187887, 0.183205, 0.17827, 0.174242}
```

In[24]:=

```
Plot[
  {funeven[h, x], pscos[x, c, 5], pscos[x, c, 10], pscos[x, c, 15]},
  {x, -Pi, Pi}, PlotStyle -> {Black, Blue, Red, {Thick, Orange}},
  PlotRange -> {{-Pi, Pi}, {0, 7}}, ImageSize -> {500, 350}]
```

Out[24]=



In[25]:=

```
d = N[fsincoef[h, 20]]
```


Out[25]=

```
{1.13933, 0.387503, 0.579453, 0.309554, 0.428448,
0.265996, 0.352804, 0.237251, 0.305833, 0.216423,
0.273203, 0.200414, 0.24891, 0.187598, 0.22995,
0.177029, 0.214638, 0.168114, 0.201947, 0.160457}
```

In[26]:=

```
Plot[
  {funodd[h, x], pssin[x, d, 10], pssin[x, d, 15], pssin[x, d, 20]},
  {x, -Pi, Pi}, PlotStyle -> {Black, Blue, Red, {Thick, Orange}},
  PlotRange -> {-7, 7}, ImageSize -> {500, 350}]
```

Out[26]=



Both plots indicate that the Fourier sine and cosine series of $h(x)$ does indeed converge to the function and in fact blow up at the singular point of h , facts we will not prove but are nonetheless, true.

Built in Functions

With the introduction of *Mathematica* 8, a variety of Fourier series functions are now built in. Here is one example, others can be seen in the documentation.

In[27]:=

```
f[x_] := Piecewise[{{-1/2, -Pi ≤ x < -Pi/2},
  {1/2, -Pi/2 ≤ x < 0}, {0, 0 ≤ x < Pi/2}, {x - Pi/2, Pi/2 ≤ x < Pi}}]
FourierTrigSeries[f[x], x, 8]
```

Out[28]=

$$\begin{aligned} & \frac{\pi}{16} + \frac{\cos[2x]}{2\pi} - \frac{4\cos[3x]}{9\pi} + \frac{4\cos[5x]}{25\pi} + \frac{\cos[6x]}{18\pi} - \\ & \frac{8\cos[7x]}{49\pi} + \frac{(-2+\pi)\sin[x]}{2\pi} - \frac{(4+\pi)\sin[2x]}{4\pi} + \\ & \frac{1}{2} \left(\frac{1}{3} + \frac{2}{9\pi} \right) \sin[3x] - \frac{1}{8} \sin[4x] + \frac{1}{2} \left(\frac{1}{5} - \frac{2}{25\pi} \right) \sin[5x] - \\ & \frac{(4+\pi)\sin[6x]}{12\pi} + \frac{1}{2} \left(\frac{1}{7} + \frac{2}{49\pi} \right) \sin[7x] - \frac{1}{16} \sin[8x] \end{aligned}$$

In[29]:=

```
Plot[{f[x], %}, {x, -Pi, Pi}]
```

Out[29]=

